

Model checking with SPIN

Software Analysis, Spring 2024

Luca Di Bello

May 18, 2024

Contents

1	Introduction	1
2	ProMeLa Model	2
2.1	Sequential Process	3
2.2	Parallel Process	3
2.2.1	Worker Process	4
2.3	Parameters: MAX and LENGTH	5
3	LTL Properties	7
3.1	Verify completion of both sequential and parallel processes	7
3.2	Same most frequent value	7
3.3	Sum of frequencies	7
3.4	Invalid LTL formula: partial result	8
4	Automatic SPIN verification via automation script	8

1 Introduction

This assignment examines the implementation of model checking using the [SPIN](#) tool to verify the correctness of two versions of a frequency counter program: one sequential and the other parallel. Model checking is a technique that allows to verify the correctness of certain properties of a system described in a finite-state model.

The following sections will discuss how the program has been modeled using [ProMeLa](#) language, which Linear Temporal Logic (LTL) properties have been defined to verify the correctness of the program, and how the verification has been performed using SPIN.

2 ProMeLa Model

The ProMeLa model consists of two main processes: the first process handles the sequential computation of frequency counts, storing results in an array `sequential_counts`. The second process on the other hand, initiates parallel computation by spawning a worker process for each possible value in the input array. These workers update an array `parallel_counts` concurrently without encountering race conditions as each worker updates a unique position in the array.

As explicitly stated in the assignment, the ProMeLa model presents two constraints:

1. **MAX**: It represents the maximum value that can be assigned to an element in the array. This constant is used as an upper bound when filling the input array with random values.
2. **LENGTH**: the length of the input array. This constant is used many times in the model to iterate over the input array while performing various operations.

The model presents an `init` block that initializes the input array with random values between 0 and **MAX** and starts both the sequential and parallel processes. The code is available in listing 1.

```

1 // Define the maximum number of elements in the array
2 #define MAX 2
3 #define LENGTH 2
4
5 // Define the variables
6 int a[LENGTH];
7
8 // Keep track of result of both versions of the program
9 int sequential_counts[MAX + 1];
10 int parallel_counts[MAX + 1];
11
12 // Entry point of the program
13 init {
14     // Initialize the array non-deterministically
15     printf("Random state:\n")
16     int i;
17     for (i : 0 .. LENGTH - 1) {
18         // Select a random value for the array
19         int v;
20         select(v : 0 .. MAX);
21         // Assign the value to the array
22         a[i] = v;
23
24         // Print the value
25         printf("\ta[%d] = %d\n", i, v);
26     }
27
28     // Run the sequential version of the program
29     printf("Running sequential version...\n");
30     run sequentialCounter();
31
32     // Run the parallel version of the program
33     printf("Running parallel version...\n");
34     run parallelCounter();
35 }

```

Listing 1: ProMeLa array initialization and start of sequential and parallel processes

From the code above is possible to see that both versions of the program have been started using the `run` keyword inside the same ProMeLa model. This is different from the Java implementation, where the two versions were implemented in separate classes. This design choice was essential to allow the

verification of both versions of the program in the same model; otherwise, it would be impossible to perform verifications that compare data yielded by two different simulations.

In the abstraction, the sequential and parallel processes are implemented as separate processes. An in-depth analysis of each process is provided in the following sections.

2.1 Sequential Process

The sequential version of the frequency counter program is implemented in the `sequentialCounter` process. This process iterates over the input array and increments the corresponding position in the `sequential_counts` array.

The ProMeLa implementation of this version of the program is almost identical to the Java implementation. This was expected, as both languages use a C-like syntax, and the logic of the program is simple. The code is available in listing 3.

```

1 public int mostFrequent() {
2     int mostFrequent = -1;
3     int maxFrequency = -1;
4     int[] frequencies = new int[max + 1];
5     for (int k = 0; k < a.length; k++)
6     {
7         int value = a[k];
7         frequencies[value] += 1;
8         int frequency = frequencies[
9         value];
9         if (frequency > maxFrequency)
10        {
11            mostFrequent = value;
11            maxFrequency = frequency;
12        }
13    }
14    return mostFrequent;
15 }
16
17

```

Listing 2: Java sequential version of the frequency counter program

```

1 proctype sequentialCounter() {
2     int maxFrequency = -1;
3     int k;
4     for (k : 0 .. LENGTH - 1) {
5         int value = a[k];
6         sequential_counts[value] =
7         sequential_counts[value] + 1;
8         if
9         :: sequential_counts[value] >
10        maxFrequency ->
11            maxFrequency =
12            sequential_counts[value];
13            sequential_result = value;
14            :: else -> skip;
15        fi
16    }
17    // Signal that the sequential
18    version is done
19    sequentialDone = 1;
20 }
21
22

```

Listing 3: ProMeLa sequential version of the frequency counter program

The main difference between the Java and ProMeLa implementations is the way the program yields the result: in the Java implementation, the result is simply returned by the method, while in the ProMeLa implementation, the result is stored in a global variable `sequential_result`. This is necessary as the process cannot return a value.

2.2 Parallel Process

The parallel version of the frequency counter program is implemented in the `parallelCounter` process. This process spawns `MAX + 1` worker processes that concurrently count the frequency of each value in the input array. Each worker process is started with a unique value to count, and by iterating over the input array, increments the corresponding position in the `parallel_counts` array.

After all worker processes have completed, the main parallel counter process iterates through the results and saves the value with the highest frequency inside the `parallel_result` variable.

To detect when all worker processes have completed, a *channel* is used to synchronize the main process with the workers. The channel is created in the main process and passed as an argument to each worker process. As soon as a worker process completes, it sends its PID through the channel.

Since we start $\text{MAX} + 1$ worker processes, we expect to read $\text{MAX} + 1$ values from the channel. By leveraging this, the *parallelCounter* process can detect when all worker processes have completed. The code is available in listing 4.

```

1 proctype parallelCounter() {
2   // Create channel to wait for workers to finish
3   chan joinCh = [MAX + 1] of { pid };
4   // Create array of PID's for workers
5   pid workers[MAX + 1];
6   // Create a worker for each possible value
7   int i;
8   for (i : 0 .. MAX) {
9     workers[i] = run parallelWorker(i, joinCh);
10  }
11  // Wait for all workers to finish by reading from the channel
12  for (i : 0 .. MAX) {
13    int done;
14    joinCh ? done;
15  }
16  // If we are here, all workers are done as we read MAX+1 values from the channel
17  printf("\t [!] all workers are done\n");
18  ...
19 }

```

Listing 4: ProMeLa parallel frequency counter - worker synchronization with channel

This implementation differs from the Java implementation, where it does not wait for all threads to complete but instead starts them all together and then waits for each thread individually to complete and save its result. In the ProMeLa abstraction, we start the worker processes right away and use a channel to wait for all of them to complete before saving the actual result.

2.2.1 Worker Process

The worker process has been implemented in the *parallelWorker* process. As mentioned in the previous section, each worker process is started with a unique value to count and increments the corresponding position in the *parallel_counts* array when it finds the value in the input array. The code is available in listing 6.

The main difference between the two implementations is how the worker process is started. In the Java implementation, this required the creation of a specific class that implements the *Runnable* interface, while in the ProMeLa implementation, this each worker is a process that is started by the main *parallelCounter* process.

```

1 protected int frequencyOf(int n) {
2     int frequency = 0;
3     for (int value: a) {
4         if (value == n)
5             frequency += 1;
6     }
7     return frequency;
8 }
9
10 class ThreadedCounter extends
    SequentialCounter implements
    Runnable
11 {
12     private int frequency;
13     private int n;
14
15     ThreadedCounter(int[] a, int n, int
        max) {
16         super(a, max);
17         this.n = n;
18     }
19
20     public void run() {
21         frequency = frequencyOf(n);
22     }
23
24     public int frequency() {
25         return frequency;
26     }
27 }
28

```

Listing 5: Java parallel worker thread implementation leveraging Threads

A notable implementation detail is that the Java worker, using an external function named *frequencyOf*, only computes the frequency of a specific value. In contrast, the ProMeLa worker process not only computes this frequency but also saves the result in the `parallel_counts` array and synchronizes with the parent process using the channel (as detailed in section 2.2).

2.3 Parameters: MAX and LENGTH

If MAX and LENGTH are too high, the model checking process can take a long time to complete, or even run out of memory. To analyze the behavior of the model checking process with different values of MAX and LENGTH, I first run the checker by keeping one parameter fixed and varying the other. This has been done for both parameters to understand their impact on the model checking performance. The results can be seen in figure 1.

```

1 // The worker process represents a
    thread that looks for the count of
    a specific value in the array
2 proctype parallelWorker(int value;
    chan out) {
3     // Look for the value in the array
4     int frequency = 0;
5     int i;
6     for (i : 0 .. LENGTH - 1) {
7         if
8             :: a[i] == value -> frequency =
                frequency + 1;
9             :: else -> skip;
10        fi
11    }
12    // Update the value in the parallel
        counts array
13    parallel_counts[value] = frequency;
14    printf("Worker for value %d is done\n", value);
15    out ! _pid;
16 }
17

```

Listing 6: ProMeLa parallel frequency counter - worker process




Figure 1: Model checking execution time with different values of MAX and LENGTH

Then, to have a better understanding, I also decided to increase the values of both MAX and LENGTH to understand how the model checking process behaves. Then, I merged the results in a single plot, available in figure 2.



Figure 2: Model checking execution time with increasing values of MAX and LENGTH

From the plot, it is possible to see that the model checking process is fast enough when the values of MAX and LENGTH are extremely low (≤ 2). However, as the values of MAX and LENGTH increase, the model checking process becomes exponentially slower. This is expected, as the model checking

process has to explore all possible states of the system, and the number of states grows exponentially with the values of MAX and LENGTH.

After the benchmarks outlined above, has been decided to set the valued of MAX and LENGTH to 2, to keep the model checking process fast and efficient enough to be able to analyze the behavior of the model.

3 LTL Properties

In the following subsections, I will discuss the LTL properties that have been defined to satisfy the requirements of the assignment. I developed

3.1 Verify completion of both sequential and parallel processes

To be able to verify the completion of both the sequential and parallel processes, I decided to use two global variables: `sequentialDone` and `parallelDone`. These variables are set to 0 by default, and set to 1 when the respective process has completed. To verify the completion of both processes, I defined the following LTL property:

```
ltl termination { <> (sequentialdone == 1 && paralleldone == 1) }
```

Listing 7: LTL property to verify the completion of both sequential and parallel processes

3.2 Same most frequent value

To ensure that both versions of the frequency counter give the same result, I defined the following LTL property:

```
ltl sameResult { [] (sequentialDone == 1 && parallelDone == 1) -> (sequential_result == parallel_result)}
```

Listing 8: LTL property to verify that both versions of the frequency counter give the same result

This LTL verifies that the result of both versions (sequential and parallel) is the same after both processes have completed. This is important property to verify, as the two algorithms are expected to yield the same result.

3.3 Sum of frequencies

As we are interested in verifying the correctness of the frequency counter program, I decided to add an additional LTL property that ensures that the sum of the frequencies of all values in the `sequential_counts` and `parallel_counts` arrays is equal to the length of the input array. This property is defined as follows:

```
ltl sumCounts { [] (sequentialDone == 1 && parallelDone == 1) -> (sumCountsSequential == LENGTH && sumCountsParallel == LENGTH) }
```

Listing 9: LTL properties to verify the sum of frequencies in the sequential and parallel arrays

As also possible to see from the code above, to verify this property I had to add two counters, `sumCountsSequential` and `sumCountsParallel`, that are computed by summing the frequencies of all values in the respective arrays right before completing the respective processes. This way, I can verify that the sum of the frequencies is equal to the length of the input array.

3.4 Invalid LTL formula: partial result

As requested by the assignment, I also added on purpose an invalid LTL formula that SPIN will not be able to verify. The property I decided to add is the following:

```
1 ltl alwaysSameResult { [] (sequential_result == parallel_result) }
```

Listing 10: Invalid LTL property that SPIN will not be able to verify

This property is very similar to the `sameResult` property (refer to section 3.2), but it does not take into account the completion of the processes. This LTL is wrong, as the two results are not expected to be the same while the processes are still running.

This is because the two versions of the program are not executed in parallel (first the sequential version and then the parallel version) and also, the two versions work in different ways. The sequential version computes the result and updates the `sequential_result` variable as soon as it finds a value with a higher frequency. The parallel version, on the other hand, spawns multiple worker processes that concurrently compute the frequency of each value in the input array, and, only after all workers have completed, the main process updates the `parallel_result` variable.

For these multiple reasons, the `alwaysSameResult` property is invalid and SPIN will find a counterexample right after the first iteration of the sequential process. This is the counterexample that SPIN will find (truncate for brevity):

```
1 #processes: 2
2         a[0] = 0
3         a[1] = 0
4         sequential_counts[0] = 1
5         sequential_counts[1] = 0
6         sequential_counts[2] = 0
7         parallel_counts[0] = 0
8         parallel_counts[1] = 0
9         parallel_counts[2] = 0
10        sequential_result = 0
11        parallel_result = -1
12        sequentialDone = 0
13        parallelDone = 0
14        sumCountsSequential = 0
15        sumCountsParallel = 0
16 41:      proc 1 (sequentialCounter:1) ../src/promela/model.pml:51 (state 12)
17 41:      proc 0 (:init::1) ../src/promela/model.pml:42 (state 22)
18 41:      proc - (notSameResult:1) _spin_nvr.tmp:60 (state 6)
```

From the output above, it is possible to understand why, where and when SPIN found the counterexample: the LTL property is violated right after the first iteration of the sequential process as the two results are different (`sequential_result = 0` and `parallel_result = -1`). At the moment of the violation, the parallel process has not even started yet.

4 Automatic SPIN verification via automation script

To automate the verification process, has been created a script that will build the model, compile the resulting analyzer using `gcc`, and run the model checker using SPIN for each of the defined LTL properties. To run the script, it is possible to use the provided Makefile target `"run"` or run the script directly. Use the commands below to run the script:

```
1 make run
2 # or
3 ./scripts/run-model.sh
```